

Relativistic Motion of a Charged Particle and Pauli Algebra

Jan Vrbik

We introduce some key formulas of special relativity and apply them to the motion of a spinless, charged point particle of unit mass, subject to the Lorentz force due to an external electromagnetic field.

■ Pauli Algebra

An element of Pauli algebra consists of a complex scalar, say A , and a three-dimensional complex vector \mathbf{a} , denoted $\mathbb{A} \equiv (A, \mathbf{a})$, which thus has eight real dimensions. In effect, this is a generalization of quaternion algebra, but with complex instead of real components. In this article, we call these elements “spinors.”

A product of two spinors is a spinor defined by

$$(A, \mathbf{a}) \otimes (B, \mathbf{b}) = (AB + \mathbf{a} \cdot \mathbf{b}, A\mathbf{a} + B\mathbf{b} + i\mathbf{a} \times \mathbf{b}), \quad (1)$$

where \cdot and \times are the dot and cross products, respectively. Note that this multiplication is associative, implying that we do not need parentheses when multiplying three or more spinors. But multiplication is not commutative (the result depends on the order of factors).

There are two important unary (single-argument) operations on spinors: the first is called a reflection (denoted \mathbb{A}^-), which changes the sign of the vector part, that is, $\mathbb{A}^- \equiv (A, -\mathbf{a})$; the second takes the complex conjugate of A and of each component of \mathbf{a} ; it is denoted \mathbb{A}^* . Finally, just for convenience, we let \mathbb{A}^+ denote the combination of both of these, that is, $(\mathbb{A}^-)^* = (\mathbb{A}^*)^-$. Note that

$$\begin{aligned} (\mathbb{A} \otimes \mathbb{B})^- &= \mathbb{B}^- \otimes \mathbb{A}^- \text{ and} \\ (\mathbb{A} \otimes \mathbb{B})^* &= \mathbb{B}^* \otimes \mathbb{A}^* \text{ but} \\ (\mathbb{A} \otimes \mathbb{B})^+ &= \mathbb{A}^+ \otimes \mathbb{B}^+, \end{aligned} \quad (2)$$

which are quite easy to verify.

The corresponding *Mathematica* routines look and work like this.

```

{A_, a_} ⊗ {B_, b_List} := {A B + a.b, A b + B a + i a × b}
{2 + 3 i, {3 - i, -2 + 4 i, 1 - 3 i}} ⊗ {-3 + i, {4 + 3 i, 2 - 3 i, 4 + i}}

{21 + i, {-32 + 19 i, 23 - 14 i, 26 + 47 i}}

{A_, a_}^- := {A, -a}
{2 + 3 i, {3 - i, -2 + 4 i, 1 - 3 i}}^-

{2 + 3 i, {-3 + i, 2 - 4 i, -1 + 3 i}}

{A_, a_}^* := {A, a} /. Complex[q_, w_] → Complex[q, -w]
{2 + 3 i, {3 - i, -2 + 4 i, 1 - 3 i}}^*

{2 - 3 i, {3 + i, -2 - 4 i, 1 + 3 i}}

```

From now on we consider a three-dimensional vector to be a special case of a spinor, meaning that \mathbf{a} is shorthand for $(0, \mathbf{a})$. We can easily compute various functions of spinors (and of three-dimensional vectors, as a special case). Thus, for example, assuming that δ is a three-dimensional vector with real components, we find

$$\begin{aligned} \exp(\delta) &= 1 + \delta + \frac{\delta \otimes \delta}{2} + \frac{\delta \otimes \delta \otimes \delta}{3!} + \frac{\delta \otimes \delta \otimes \delta \otimes \delta}{4!} + \dots = \\ &\left(1 + \frac{\delta^2}{2} + \frac{\delta^4}{4!} + \dots\right) + \hat{\delta} \left(\delta + \frac{\delta^3}{3!} + \frac{\delta^5}{5!} + \dots\right) = \cosh(\delta) + \hat{\delta} \sinh(\delta), \end{aligned} \quad (3)$$

where δ is the length of δ and $\hat{\delta} \equiv \frac{\delta}{\delta}$ is a unit vector in the same direction.

Similarly, for a three-dimensional vector with pure imaginary components (which we express in the form $i\alpha$ to keep the elements of α real), the same kind of expansion yields

$$\begin{aligned} \exp(i\alpha) &= 1 + i\alpha - \frac{\alpha \otimes \alpha}{2} - i \frac{\alpha \otimes \alpha \otimes \alpha}{3!} + \frac{\alpha \otimes \alpha \otimes \alpha \otimes \alpha}{4!} + \dots = \\ &\left(1 - \frac{\alpha^2}{2} + \frac{\alpha^4}{4!} + \dots\right) + i \hat{\alpha} \left(\alpha - \frac{\alpha^3}{3!} + \frac{\alpha^5}{5!} + \dots\right) = \cos(\alpha) + i \hat{\alpha} \sin(\alpha), \end{aligned} \quad (4)$$

again where α is the length of α and $\hat{\alpha} \equiv \frac{\alpha}{\alpha}$ is a unit vector in the same direction.

This can be easily extended to a complex-vector argument, as follows.

SPexp[**a_**] := **Module** [{ $\alpha = \text{Total}[\mathbf{a}^2]^{1/2}$ }, { **Cosh**[α], $\mathbf{a} / \alpha \text{Sinh}[\alpha]$ }]

SPexp[{2, -3, 2}]

{ **Cosh**[$\sqrt{17}$], { $\frac{2 \text{Sinh}[\sqrt{17}]}{\sqrt{17}}$, $-\frac{3 \text{Sinh}[\sqrt{17}]}{\sqrt{17}}$, $\frac{2 \text{Sinh}[\sqrt{17}]}{\sqrt{17}}$ } } }

SPexp[{-i, 4 i, i}]

{ **Cos**[$3\sqrt{2}$], { $-\frac{i \text{Sin}[3\sqrt{2}]}{3\sqrt{2}}$, $\frac{2}{3} i \sqrt{2} \text{Sin}[3\sqrt{2}]$, $\frac{i \text{Sin}[3\sqrt{2}]}{3\sqrt{2}}$ } } }

SPexp[{2 - i, -3 + 4 i, 2 + i}] // **N**

{ -13.7409 + 5.72248 i, { -6.77051 + 0.363558 i,
13.1775 - 7.49763 i, -4.35315 - 5.19827 i } }

■ Special Relativity

The basic idea of Einstein's theory is to unite space and time into a single entity of space-time, whose "points" (or events) can then be represented by *real* spinors of type $X \equiv (t, \mathbf{r})$. The separation between any two such events constitutes a so-called 4-vector $(t_2 - t_1, \mathbf{r}_2 - \mathbf{r}_1)$ that, when multiplied (in the spinor sense) by its own reflection, yields a pure scalar $(t_2 - t_1)^2 - (\mathbf{r}_2 - \mathbf{r}_1) \cdot (\mathbf{r}_2 - \mathbf{r}_1)$. It is now postulated that this quantity must be invariant (having the same value) in all inertial coordinate systems (i.e. those that differ from each other by a fixed rotation and/or a boost—a motion at constant velocity). For simplicity, our choice of units sets the speed of light (which must be the same in all inertial systems) equal to 1.

The question is: how do we transform 4-vectors from one inertial frame to another, while maintaining this invariance? The answer is provided by

$$\mathbb{D}' = \mathbb{L}^* \otimes \mathbb{D} \otimes \mathbb{L}, \quad (5)$$

where \mathbb{D} and \mathbb{D}' represent a 4-vector in the old and new frame of reference, respectively, and \mathbb{L} is a spinor such that $\mathbb{L} \otimes \mathbb{L}^- \equiv \mathbb{L}^- \otimes \mathbb{L} = 1$. It is obvious that \mathbb{D}' remains real whenever \mathbb{D} is real, and that

$$\begin{aligned} S' &= (\mathbb{D}')^- \otimes \mathbb{D}' = \mathbb{L}^- \otimes \mathbb{D}^- \otimes \mathbb{L}^+ \otimes \mathbb{L}^* \otimes \mathbb{D} \otimes \mathbb{L} = \\ &= \mathbb{L}^- \otimes \mathbb{D}^- \otimes (\mathbb{L} \otimes \mathbb{L}^-)^* \otimes \mathbb{D} \otimes \mathbb{L} = \mathbb{L}^- \otimes S \otimes \mathbb{L} = S, \end{aligned} \quad (6)$$

where $S = \mathbb{D}^- \otimes \mathbb{D}$ (the scalar invariant of \mathbb{D}). This shows that S has the same value in all inertial frames of reference. Requiring reflection to be a frame-independent operation (meaning that $(\mathbb{D}^-)' \equiv (\mathbb{D}')^-$ for every 4-vector) leads to

$$(\mathbb{D}^-)' = (\mathbb{L}^* \otimes \mathbb{D} \otimes \mathbb{L})^- = \mathbb{L}^- \otimes \mathbb{D}^- \otimes \mathbb{L}^+. \quad (7)$$

From this, we can see that \mathbb{D}^- transforms differently from \mathbb{D} , being an example of a four-dimensional *covector* (a 4⁻-vector, in our notation). Another important example of a 4⁻-vector is the (∂_t, ∇) operator, where ∇ stands for the usual three-dimensional (spatial) gradient (the collection of x , y , and z partial derivatives).

One can show that $\mathbb{L}^- \otimes \mathbb{L} = 1$ if and only if $\mathbb{L} = \exp\left(\frac{\mathbf{a}}{2}\right)$, where \mathbf{a} is a “pure” vector (i.e. three dimensional, but potentially complex valued). This is a fully general but rather inconvenient form of \mathbb{L} ; fortunately, one can prove that any such \mathbb{L} can be expressed as a product of an ordinary three-dimensional rotation $\exp\left(i \frac{\alpha}{2}\right)$ and a boost $\exp\left(\frac{\delta}{2}\right)$, where α and δ are real-valued vectors. To see that $\exp\left(i \frac{\alpha}{2}\right)$ results in an ordinary rotation of the vector part of \mathbb{D} (where the magnitude of α specifies the angle and the direction of α defines the axis of rotation), consider

$$\begin{aligned} \exp\left(-i \frac{\alpha}{2}\right) \otimes \mathbb{D} \otimes \exp\left(i \frac{\alpha}{2}\right) = \\ (D, \mathbf{d}_{\parallel}) + \exp(-i\alpha) \otimes \mathbf{d}_{\perp} = (D, \mathbf{d}_{\parallel} + \cos(\alpha) \mathbf{d}_{\perp} + \sin(\alpha) \hat{\alpha} \times \mathbf{d}_{\perp}), \end{aligned} \quad (8)$$

where $\mathbf{d} = \mathbf{d}_{\parallel} + \mathbf{d}_{\perp}$ is the decomposition of \mathbf{d} into components parallel and perpendicular to α , respectively. Note that $(D, \mathbf{d}_{\parallel})$ commutes with α (and any function thereof); consequently, it remains unchanged by this transformation. On the other hand, \mathbf{d}_{\perp} and α anti-commute (i.e. $\alpha \otimes \mathbf{d}_{\perp} = -\mathbf{d}_{\perp} \otimes \alpha$), implying

$$\mathbf{d}_{\perp} \otimes \exp\left(i \frac{\alpha}{2}\right) = \exp\left(-i \frac{\alpha}{2}\right) \otimes \mathbf{d}_{\perp} \quad (9)$$

and, consequently, the rest of (8). One can thus see that this transformation leaves the scalar part of a 4-vector intact, and rotates the vector part using $\hat{\alpha}$ and α as the axis and angle of rotation (of the old frame, with respect to the new one), respectively.

Similarly, using $\mathbb{L} = \exp\left(\frac{\delta}{2}\right)$, we get

$$\begin{aligned} \exp\left(\frac{\delta}{2}\right) \otimes \mathbb{D} \otimes \exp\left(\frac{\delta}{2}\right) = \exp(\delta) \otimes (D, \mathbf{d}_{\parallel}) + \mathbf{d}_{\perp} = \\ (D \cosh(\delta) + \mathbf{d}_{\parallel} \cdot \hat{\delta} \sinh(\delta), D \sinh(\delta) \hat{\delta} + \cosh(\delta) \mathbf{d}_{\parallel} + \mathbf{d}_{\perp}). \end{aligned} \quad (10)$$

Here, δ can be replaced by $\hat{\mathbf{v}} \operatorname{arctanh} v$, where $\hat{\mathbf{v}}$ and $0 \leq v \leq 1$ represent the unit direction and the magnitude of a three-dimensional vector (a velocity of the old frame with respect to the new one), respectively. This leads to the following simplified (and physically more meaningful) result of such a boost, applied to \mathbb{D} :

$$\left(\frac{D + \mathbf{d}_{\parallel} \cdot \mathbf{v}}{\sqrt{1 - v^2}}, \frac{\mathbf{d}_{\parallel} + D \mathbf{v}}{\sqrt{1 - v^2}} + \mathbf{d}_{\perp} \right). \quad (11)$$

□ Electromagnetic Fields

A 4-vector field of central significance is the electromagnetic potential (φ, \mathbf{A}) , where φ and each component of \mathbf{A} are real-valued functions of a spacetime location (i.e. of x , y , z , and t). Now comes an important point: within the current framework, it is physically meaningless to multiply two 4-vectors (we would not know how to transform such a product from one inertial frame to another), but it is possible to multiply a 4⁻-vector by a 4-vector, creating what we call a mixed vector. It is thus quite legitimate to do

$$(\partial_t, \nabla) \otimes (\varphi, \mathbf{A}) = (\partial_t \varphi + \nabla \cdot \mathbf{A}, \partial_t \mathbf{A} + \nabla \varphi + i \nabla \times \mathbf{A}), \quad (12)$$

creating a mixed vector, which has its own new way of transforming (shown shortly).

To physically interpret the right-hand side of (12), we first impose the so-called Lorenz condition (often misattributed to the more famous Lorentz) of $\partial_t \varphi + \nabla \cdot \mathbf{A} \equiv 0$ (we prove shortly that this condition is frame invariant), and identify $\partial_t \mathbf{A} + \nabla \varphi + i \nabla \times \mathbf{A}$ with $-\mathbf{E} + i \mathbf{B}$, where \mathbf{E} and \mathbf{B} are the resulting (real-valued) electric and magnetic fields, respectively.

One can now show rather easily that $\mathbb{L} = \exp(i \frac{\alpha}{2})$ will simply rotate the two fields. On the other hand, a boost $\mathbb{L} = \exp(\frac{\delta}{2})$ results in

$$\begin{aligned} \exp\left(-\frac{\delta}{2}\right) \otimes (0, -\mathbf{E} + i \mathbf{B}) \otimes \exp\left(\frac{\delta}{2}\right) = \\ \exp(-\delta) \otimes (0, -\mathbf{E}_{\perp} + i \mathbf{B}_{\perp}) - \mathbf{E}_{\parallel} + i \mathbf{B}_{\parallel} = \\ \frac{-\mathbf{E}_{\perp} + \mathbf{v} \times \mathbf{B}_{\perp} + i \mathbf{B}_{\perp} + i \mathbf{v} \times \mathbf{E}_{\perp}}{\sqrt{1 - v^2}} - \mathbf{E}_{\parallel} + i \mathbf{B}_{\parallel}, \end{aligned} \quad (13)$$

since transforming a mixed vector \mathbb{M} is done by $\mathbb{M}' = \mathbb{L}^{-1} \otimes \mathbb{M} \otimes \mathbb{L}$. In both cases (rotation and boost), the scalar part remains equal to zero, thus proving invariance of the Lorenz condition.

Unlike 4-vectors, mixed vectors can be multiplied, yielding a new mixed vector (in terms of its transformation properties). Thus, for example,

$$(0, -\mathbf{E} + i \mathbf{B}) \otimes (0, -\mathbf{E} + i \mathbf{B}) = (\mathbf{E}^2 - \mathbf{B}^2 - 2i \mathbf{E} \cdot \mathbf{B}, \mathbf{0}) \quad (14)$$

tells us that both $\mathbf{E}^2 - \mathbf{B}^2$ and $\mathbf{E} \cdot \mathbf{B}$ are invariant scalar fields.

Similarly, multiplying a 4-vector by a mixed vector returns a 4-vector; for example,

$$(\partial_t, -\nabla) \otimes (0, -\mathbf{E} + i \mathbf{B}) = (\nabla \cdot \mathbf{E} - i \nabla \cdot \mathbf{B}, -\partial_t \mathbf{E} + i \partial_t \mathbf{B} + i \nabla \times \mathbf{E} + \nabla \times \mathbf{B}). \quad (15)$$

This time, the result must equal the charge/current density (yet another basic 4-vector of the theory); this results in a compact, Dirac-algebra formulation of the usual Maxwell equations. As an example, the following program computes the electromagnetic field of a point-like massive particle of a unit charge, moving at a uniform speed v along the z axis; it also verifies that the result satisfies Maxwell's equations.

```
EM = {0, {x, y, z} / (x^2 + y^2 + z^2)^(3/2)};
Lb = SPexp[{0, 0, ArcTanh[v] / 2}];
EM = (Lb^-1 \otimes EM) \otimes Lb // PowerExpand // Simplify;
EM = EM /. z -> (z - t v) / Sqrt[1 - v^2]

{0, {
  (x + i v y) / (Sqrt[1 - v^2] (x^2 + y^2 + (z - t v)^2 / (1 - v^2))^(3/2)),
  (-i v x + y) / (Sqrt[1 - v^2] (x^2 + y^2 + (z - t v)^2 / (1 - v^2))^(3/2)),
  (-t v + z) / (Sqrt[1 - v^2] (x^2 + y^2 + (z - t v)^2 / (1 - v^2))^(3/2))
}}

{dt, -{dx, dy, dz}} \otimes EM /. di_ q_ -> D[q, i] // Simplify

{0, {0, 0, 0}}
```

■ Dynamics of a Point-Like Charged Particle

We choose units in which both the rest mass and the charge of the particle equal 1. Its instantaneous location is denoted $\mathbb{X} \equiv (t, \mathbf{r}$, where the three components of \mathbf{r} are functions of t . To make the subsequent equations frame independent (having the same form in all inertial frames), we now need to introduce the particle's so-called proper time τ by

$$\frac{d\tau}{dt} = \sqrt{1 - \left(\frac{d\mathbf{r}}{dt}\right)^2}. \quad (16)$$

We already know that $(dt, d\mathbf{r})$ transforms as a 4-vector; $(dt, -d\mathbf{r}) \otimes (dt, d\mathbf{r}) = (dt^2 - d\mathbf{r}^2, \mathbf{0})$, thus $dt^2 - d\mathbf{r}^2$ and, consequently, $dt \sqrt{1 - \dot{\mathbf{r}}^2}$ are relativistically invariant (the dot over \mathbf{r} implies differentiation with respect to t). Proper time τ is then computed by the corresponding integral, namely

$$\tau \equiv \int \sqrt{1 - \dot{\mathbf{r}}^2} dt, \quad (17)$$

which, being a “sum” of invariant quantities, is an invariant scalar as well. This implies that

$$\mathbb{P} \equiv \frac{d\mathbb{X}}{d\tau} = \left(\frac{1}{\sqrt{1 - \dot{\mathbf{r}}^2}}, \frac{\dot{\mathbf{r}}}{\sqrt{1 - \dot{\mathbf{r}}^2}} \right) \quad (18)$$

transforms as a 4-vector (we call it the 4-momentum of the moving particle).

The motion of the charged particle in a given electromagnetic field can now be established by solving

$$\frac{d\mathbb{P}}{d\tau} = \text{Re}[\mathbb{P} \otimes (0, \mathbf{E} - i\mathbf{B})], \quad (19)$$

where the right-hand side is the so-called Lorentz force, and “Re” takes the real part of its argument (also a 4-vector). Note that “Re” applied to a 4-vector remains a 4-vector (since the complex conjugate of a 4-vector transforms as a 4-vector, as one can readily verify).

Based on (18), the right-hand side of (19) equals

$$\left(\frac{\dot{\mathbf{r}} \cdot \mathbf{E}}{\sqrt{1 - \dot{\mathbf{r}}^2}}, \frac{\mathbf{E} + \dot{\mathbf{r}} \times \mathbf{B}}{\sqrt{1 - \dot{\mathbf{r}}^2}} \right). \quad (20)$$

For a particle with a negative unit charge, the sign of this 4-vector would be reversed.

One way of solving the resulting 4-vector equation is to expand the left-hand side of (19):

$$\frac{d\mathbb{P}}{d\tau} = \frac{1}{\sqrt{1 - \dot{\mathbf{r}}^2}} \frac{d\mathbb{P}}{dt} = \frac{1}{\sqrt{1 - \dot{\mathbf{r}}^2}} \left(\frac{\ddot{\mathbf{r}} \cdot \dot{\mathbf{r}}}{(1 - \dot{\mathbf{r}}^2)^{3/2}}, \frac{\ddot{\mathbf{r}}(1 - \dot{\mathbf{r}}^2) + (\ddot{\mathbf{r}} \cdot \dot{\mathbf{r}})\dot{\mathbf{r}}}{(1 - \dot{\mathbf{r}}^2)^{3/2}} \right), \quad (21)$$

which has to equal (20). Canceling the scalar factor of $\frac{1}{\sqrt{1 - \dot{\mathbf{r}}^2}}$, we get

$$\left(\frac{\ddot{\mathbf{r}} \cdot \dot{\mathbf{r}}}{(1 - \dot{\mathbf{r}}^2)^{3/2}}, \frac{\ddot{\mathbf{r}}(1 - \dot{\mathbf{r}}^2) + (\ddot{\mathbf{r}} \cdot \dot{\mathbf{r}})\dot{\mathbf{r}}}{(1 - \dot{\mathbf{r}}^2)^{3/2}} \right) = (\dot{\mathbf{r}} \cdot \mathbf{E}, \mathbf{E} + \dot{\mathbf{r}} \times \mathbf{B}). \quad (22)$$

Finally, the first (scalar part) of these four equations can be obtained by taking the dot product of the remaining three equations (the vector part) and $\dot{\mathbf{r}}$, and is therefore redundant. All we need to solve in the end is

$$\frac{\ddot{\mathbf{r}}(1 - \dot{\mathbf{r}}^2) + (\ddot{\mathbf{r}} \cdot \dot{\mathbf{r}}) \dot{\mathbf{r}}}{(1 - \dot{\mathbf{r}}^2)^{3/2}} = \mathbf{E} + \dot{\mathbf{r}} \times \mathbf{B} \quad (23)$$

or, equivalently (by solving for $\ddot{\mathbf{r}}$),


$$\ddot{\mathbf{r}} = \sqrt{1 - \dot{\mathbf{r}}^2} [\mathbf{E} + \dot{\mathbf{r}} \times \mathbf{B} - (\mathbf{E} \cdot \dot{\mathbf{r}}) \dot{\mathbf{r}}], \quad (24)$$

the correctness of which can be easily verified by substituting this $\ddot{\mathbf{r}}$ into the left-hand side of (23) and obtaining the right-hand side. These equations can, in general, be solved only numerically. The resulting \mathbf{r} will be a three-component function of a frame-dependent time t (proper time τ was used only as an intermediate tool to assure agreement between inertial frames of reference; our solution remains correct in any other such frame, after the corresponding transformation). The following program finds the motion of a negatively charged particle (chosen because an attractive force makes the resulting path somehow more interesting, compared to the same-charge repulsion) in the field of the previous section.

```

E1 = - ((EM + EM*) [[2]] / 2 // Simplify) /.
  {x → x[t], y → y[t], z → z[t]};
B = ((EM - EM*) [[2]] / 2 / i // Simplify) /.
  {x → x[t], y → y[t], z → z[t]};
aux = {x'[t], y'[t], z'[t]};
eqs =
  Thread[{x''[t], y''[t], z''[t]} ==
    (E1 + aux × B - (E1.aux) aux) / Sqrt[1 - aux.aux] /. v → .5;
eqs = Flatten[{eqs, x[0] == 0, y[0] == 2, z[0] == 2, x'[0] == 0,
  y'[0] == 0, z'[0] == 0}];
sol = NDSolve[eqs, {x, y, z}, {t, 0, 8}];
ParametricPlot[{y[t], z[t]} /. sol, {t, 0, 8},
  ImageSize → 400]

```

The originally stationary unit-mass particle has been “captured” by the moving massive particle (which is assumed to be so heavy that its own motion remains unaffected), and will continue orbiting it (while following its uniform motion). It is important to realize that this formulation of the problem has ignored the fact that a moving particle generates (and radiates) an electromagnetic field of its own, which would further modify its motion. More importantly, we also know that, at an atomic level, the world is governed by quantum mechanics, ultimately resulting in a totally different description of an orbiting particle. This is the reason why the last solution is only of mathematical interest. Under a wide range of initial conditions, the light particle would be drawn to collide with the heavy one.

□ Motion in a Constant Electromagnetic Field

Things become easier when \mathbf{E} and \mathbf{B} are both constant fields (in space and time). One can then express \mathbb{P} as $\mathbb{N}^* \otimes \mathbb{P}_0 \otimes \mathbb{N}$, where \mathbb{P}_0 is the initial value of the particle’s 4-momentum, and \mathbb{N} is a spinor analogous to \mathbb{L} that (instead of transforming 4-vectors between inertial frames) advances \mathbb{P} to its current location. This time we find it more convenient to make both \mathbb{P} and \mathbb{N} functions of proper time τ . Also, one must not forget to meet the $\mathbb{P}_0^- \otimes \mathbb{P}_0 = 1$ condition, which is then automatically maintained by \mathbb{P} at all future times.

Expanding the right-hand side of (19) yields

$$\begin{aligned} & \frac{1}{2} \mathbf{P} \otimes (0, \mathbf{E} - i \mathbf{B}) + (0, \mathbf{E} + i \mathbf{B}) \otimes \mathbf{P}^* = \\ & \frac{1}{2} \mathbf{N}^* \otimes \mathbf{P}_0 \otimes \mathbf{N} \otimes (0, \mathbf{E} - i \mathbf{B}) + (0, \mathbf{E} + i \mathbf{B}) \otimes \mathbf{N}^* \otimes \mathbf{P}_0 \otimes \mathbf{N}. \end{aligned} \quad (25)$$

This should equal the left-hand side of (19), namely

$$\mathbf{N}^* \otimes \mathbf{P}_0 \otimes \frac{d\mathbf{N}}{d\tau} + \frac{d\mathbf{N}^*}{d\tau} \otimes \mathbf{P}_0 \otimes \mathbf{N}, \quad (26)$$

which implies that

$$\frac{d\mathbf{N}}{d\tau} = \frac{\mathbf{N} \otimes (0, \mathbf{E} - i \mathbf{B})}{2}, \quad (27)$$

having the obvious solution


$$\mathbf{N} = \exp\left(\frac{\tau}{2} (\mathbf{E} - i \mathbf{B})\right); \quad (28)$$

a function of a mixed vector transforms as a mixed vector. After computing \mathbf{N} and, consequently, $\mathbf{N}^* \otimes \mathbf{P}_0 \otimes \mathbf{N}$, all we need to do is to integrate this last expression in terms of τ to get a solution for (\mathbf{r}, t) . A simple example follows.

```
EM = {5.8 i, 0.5, 1. + 0.01 i}; P0 = {2.07, {0.1, .9, 0.5}};
Ns = SPexp[τ / 2 EM] // PowerExpand;
P = (Ns* ⊗ P0) ⊗ Ns // ComplexExpand;
X = Integrate[P, τ] // TrigReduce // Chop

{-0.0293244 Cos[5.69123 τ] + 58.5293 Cosh[0.00175709 τ] -
 0.0344907 Sin[5.69123 τ] + 1289.8 Sinh[0.00175709 τ],
 {0.000398497 Cos[5.69123 τ] + 1265.6 Cosh[0.00175709 τ] -
 0.000101116 Sin[5.69123 τ] + 57.2398 Sinh[0.00175709 τ],
 0.0575475 Cos[5.69123 τ] - 10.0575 Cosh[0.00175709 τ] +
 0.226794 Sin[5.69123 τ] - 222.377 Sinh[0.00175709 τ],
 -0.225068 Cos[5.69123 τ] + 7.29507 Cosh[0.00175709 τ] +
 0.0534948 Sin[5.69123 τ] + 111.291 Sinh[0.00175709 τ]}
```

```
ParametricPlot3D[X[[2]], {τ, 0, 3}]
```



In this example, we have been able to obtain an explicit analytic solution.

In most textbooks (e.g., [1], [2]), the treatment we have described is done using conventional tensor analysis, with covariant and contravariant vectors. It is always illuminating, however, to obtain the same results using a completely different approach, such as the one favored by Hestenes [3]. We have shown that *Mathematica* is capable of handling computations in any of several formalisms.

■ References

- [1] H. Goldstein, *Classical Mechanics*, 2nd ed., Reading, MA: Addison-Wesley Publishing Company, 1980.
- [2] J. D. Jackson, *Classical Electrodynamics*, New York: John Wiley, 1962.
- [3] D. Hestenes, *New Foundations for Classical Mechanics*, 2nd ed., Dordrecht/Boston: Kluwer Academic Publishers, 1999.

J. Vrbik, "Relativistic Motion of a Charged Particle and Pauli Algebra," *The Mathematica Journal*, 2012. [dx.doi.org/doi:10.3888/tmj.14-10](https://doi.org/10.3888/tmj.14-10).

About the Author

Jan Vrbik
 Department of Mathematics, Brock University
 500 Glenridge Ave., St. Catharines
 Ontario, Canada, L2S 3A1
jvr bik@brocku.ca